# Quintessential Maple XII

Jim Emery

11/11/2008

# Contents

# Chapter 1

# Running Maple

The Maple program can be found on the network of Silicon Graphics Machines. First login or telnet to an SGI workstation. To start maple, type the command **xmaple**. This runs the XWindows version. Type: **maple** for the command line version. For help, type **help**, or type **help(topic)**. Always terminate a command with ";" or ":". Type **quit;** to leave maple.

## 1.1 Capturing Output

When reading in a file with read(filename), where filename contains Maple commands, first type the command

```
interface(echo=2);
```

This causes both the maple input and the result to be printed. To export the session as a Latex file, we select **export latex** from the xmaple menu. An editor and or **grep** can be used to change some of the Maple latex macro commands. For example we can change the string "maplettyout" to "verbatim." We can remove lines containing the string "maplelatex," et cetera.

We can save the session to a postscript file, we can export the session as a text file, or we can export the session as a Latex file. To print a postscript file from SGI, we can simply drag its icon to a printer. Or we can issue a print command from the unix command line as in:

```
$ lp -dexocom session.ps
```

This command sends the file to the printer called **exocom**. Suppose we want to extract the commands that we issued in the session to a file. We can use the unix command **grep**, which outputs those lines of a file that contain a specified string. For example, because maple commands start with the "less than" character, so we can instruct **grep** to write those lines of a file that start with "less than" to a file called out:

```
grep '^<' session.txt > out
```

The single quotes are used to prevent the command line interpreter from becoming confused. The "caret" character matches the beginning of a line. Now we can edit the new file, which is named "out". Then we can start Maple and read this command file, using the maple read() command. This is a good way to work with maple, because we can edit the command file until we can get to perform correctly in Maple, with no errors.

The maple latex output assumes the existence of a latex style macro constructed for maple, which interprets commands like "maplettyout" to start a special kind of formating. We can cleanup the file and replace these special commands by using the following unix script:

```
% cat ~/com/maplecln
grep -v maplesepline $1  > out1
grep -v maplelatex out1 > out2
cs2s out2 out1 mapleinput verbatim
cs2s out1 $2 maplettyout verbatim
```

We use grep with the "v" parameter to write lines that do not contain a string, to eliminate lines containing a string such as "maplesepline" which is a macro to draw a separating line. The program cs2s is used to replace a string such as "mapleinput" with the string "verbatim". In Latex lines between "verbatim" commands are not interpreted and are printed as is in a typewriter font.

## 1.2 Distinguishing Formulas And Functions

# Chapter 2

# A Potpouri of Maple Commands

## 2.1 Introduction

Here we will give a set of Maple commands to illustrate the capability of Maple.

```
Maple Computer Algebra Commands

 (A percentage sign represents the previous expression, in Maple V a " was used fo
 restart;
 f:=sin(x);
 diff(f,x);
 int(f,x);
 taylor(f,x=0,5);
 print(a);
 A:=array([a,b],[c,d]);
 linsolve(A,B);
 op(2,e);
 expand(a);
 factor(e);
 quit
 normal(e);
 collect(e,x);
```

```
radsimp(e);
Solve({e1=0,e2=0},{s,t});
read file;
numer(e);
denom(e);
quo(c,d,x,'r');
coef(e,x,3);
subs(t=x+sin(x),e);
aa:= proc(n,k,x) local vp;
 ......
end;
# comment
evalf(b);
E (exponential e)
with(linalg);
 a:=array(1..1,1..5,[ [1],[2],[3],[4],[5] ]);
c=evalm(a &* b); (matrix multiplication)
for i from 0 to 5 do .... od;
readlib(fortran)
fortran(a);
latex(a);
convert(g,polynom);
ifactor(10!);
plot(sin(x)*exp(-x/3),x=0..2*PI);
resultant(e1,e2,u);
if k > 1 then v:= n*(n-1)/2 fi;
```

# Chapter 3

# Examples

## 3.1 Numerical Evaluation of Formulas

Maple can be used as a calculator for evaluating formulas and equations. The static deformation of a thin plate is given by the biharmonic equation

$$\nabla^4 w(x, y) = -\frac{p}{D},$$

where $D$ is the plate stiffness, and $p$ is the pressure. Values for $D$ can be computed in an interactive Maple session. The formula for $D$ is

$$D = \frac{h^3 E}{12(1 - \nu^2)}.$$

After the equation and values have been entered, we can move the mouse to a formula defining one of the variables, and make a change. Hitting the return key then causes a the formula to be reevaluated. This must also be done for any formula that is affected by any change to a variable, or a dependent expression. That is, suppose we change Poisson's ratio to $.345O$. Then the formula for $D$, the plate stiffness should change. To make it change, move the mouse to the formula for $D$, and hit the return key.

```
#plate stiffness;

'd=h^3*E0/(12*(1- nu^2))';
```

$$d = \frac{h^3 \, E0}{12 - 12\,\nu^2}$$

```
E0:=200E9;#Young's Modulus (Pascals);
```

$$E0 := .200\,10^{12}$$

```
h:=5E-3;#plate thickness (meters);
```

$$h := .005$$

```
nu:=.355;#Poisson's Ratio;
```

$$\nu := .355$$

```
d:=h^3*E0/(12*(1- nu^2));#(Nt Meters);
```

$$d := 2383.744768$$

## 3.2 The Volume of a Spherical Cap

The volume may be considered as a stack of disks of thickness $dy$, with $y$ varying from $a$ to $r$. Thus

```
> v(a,r)=int(pi*x(y)^2,y=a..r);
```

$$\mathrm{v}(\,a, r\,) = \int_a^r \pi\,\mathrm{x}(\,y\,)^2\,dy$$

```
> subs(x(y) = sqrt(r^2 - y^2),");
```

$$\mathrm{v}(\,a, r\,) = \int_a^r \pi\,(\,r^2 - y^2\,)\,dy$$

```
> eval(");
```

$$\mathrm{v}(a, r) = \frac{2}{3}\ \pi\, r^3 - \frac{1}{3}\ \pi\, a\,(3\, r^2 - \ a^2)$$

```
> factor(");
```

$$\mathrm{v}(a, r) = \frac{1}{3}\ \pi\,(2\, r + a)\,(r - a)^2$$

```
v(a,r)=int(Pi*x(y)^2,y=a..r);
```

$$\mathrm{v}(a, r) = \int_a^r \pi\, \mathrm{x}(y)^2\, dy$$

```
subs(x(y) = sqrt(r^2 - y^2),");
```

$$\mathrm{v}(a, r) = \int_a^r \pi\,(r^2 - y^2)\, dy$$

```
eval(");
```

$$\mathrm{v}(a, r) = \frac{2}{3}\ \pi\, r^3 - \frac{1}{3}\ \pi\, a\,(3\, r^2 - \ a^2)$$

```
factor(");
```

$$\mathrm{v}(a, r) = \frac{1}{3}\ \pi\,(2\, r + a)\,(r - a)^2$$

```
e1:=";
```

$$\mathit{e1} := \mathrm{v}(a, r) = \frac{1}{3}\ \pi\,(2\, r + a)\,(r - a)^2$$

```
evalf(subs({a=.123,r=8.56},e1));
```

$$\mathrm{v}(.123, 8.56) = 1285.338583$$

Let $b = r - a$ be the depth of a spherical cap of radius r. Then its volume is

$$V = \frac{\pi}{3}(3r - b)b^2.$$




Figure 3.1: A Tank Consisting of A Cylinder and Two Spherical End Caps.

## 3.3 Numerical Evaluation of Formulas

Suppose we wish to compute the volume of a tank which consists of a cylinder with two spherical end caps, not necessarily hemespheres. Suppose the radius of the cylinder is $c$, the length of the cylinder is $2h$, the radius of each spherical cap is $r$ and the depth of each spherical cap is $b$, where

$$b = r - a,$$

and

$$a = \sqrt{r^2 - c^2}.$$

See the figure. Then the volume of the tank is

$$\frac{2\pi}{3}(3r - b)b^2 + 2h\pi c^2.$$

We may compute numerical values with maple using the following commands:

```
v:=((2*Pi)/3) *(3*r-b)*b^2 + 2*h*Pi*c^2;
```

$$v := \frac{2}{3}\,\pi\,(\,3\,r - b\,)\,b^2 + 2\,h\,\pi\,c^2$$

```
b:=r-sqrt(r^2-c^2);
```

$$b := r - \sqrt{r^2 - c^2}$$

```
'v'=evalf(subs({r=1.,c=.4,h=.2},v));
```

$$v = .2436353269$$

```
'b' = evalf(subs({r=1.,c=.4,h=.2},b));
```

$$b = .0834848610$$

The command evalf evaluates the expression as a floating point number. One can change the values and click on a statement with the mouse to reevaluate it. Warning, in the interactive session, when one clicks on a statement to reevaluate it, if following statements depend on that statement, then they to must be clicked on and reevaluated, otherwise they may no longer be valid.

## 3.4 Eliminating Variables: the Resultant

Let a circle have radius $r$ and center $(r, 0)$. Let the line with slope $t$,

$$y = tx$$

intersect the circle at $(x, y)$. The circle has equation

$$(x - r)^2 + y^2 = r^2.$$

Solving these two equations

$$x = \frac{2r}{1 + t^2}$$

$$y = \frac{2tr}{1 + t^2}$$

Translating the circle so that it has center $(h, k)$ we find

$$x = \frac{r + h + t^2(h - r)}{1 + t^2}$$

and

$$y = \frac{2rt + k + kt^2}{1 + t^2}.$$

These are rational parametric equations for a circle. They do not involve sines and cosines. When a curve or a surface have parametric representations like this, we sometimes want an implicit representation. To get an implicit representation we can use the resultant to illiminate variables. This might involve the calculation of a determinant of a matrix where the elements of the matrix are polynomial expressions. This is a difficult calculation to carry out by hand. As an example let us find the implicit representation for the circle from the rational representation. So from the rational representation we have

$$(1 + t^2)x - 2t = 0$$

$$y - tx = 0$$

Here we can easily eliminate $t$ from the two variables by hand, but for general polynomials this is often very difficult. So here is an example of using the resultant to do this:

```
> e1:= (1+t*t)*x - 2*r;

                                          2
                           e1 := (1 + t ) x - 2 r

> e2:= y-t*x ;

                               e2 := y - t x

> resultant(e1,e2,t);
```

```
                      3       2       2
                     x  - 2 x  r + y  x
```

```
> %/x;
```

```
                      3       2       2
                     x  - 2 x  r + y  x
                     ------------------
                             x
```

```
> normal(%);
```

```
                       2             2
                      x  - 2 x r + y
```

```
> resultant(e1,e2,t);
```

```
                      3       2       2
                     x  - 2 x  r + y  x
```

```
> factor(%);
```

```
                        2             2
                    x (x  - 2 x r + y )
```

There is often an extra factor introduced by the resultant that must be divided out. So the implicit equation is

$$x^2 - 2xr + y^2 = 0,$$

which is what we started with above.

## 3.5 The Cubic Hermite Interpolating Polynomial

Given a function $f(u)$, there is a unique cubic polynomial having values $f(0)$ at 0, $f(1)$ at 1, and derivatives $f'(0)$ at 0 and $f'(1)$ at 1. Let the polynomial be

$$g(u) = a_0 + a_1 u + a_2 u^2 + a_3 u^3.$$

We shall show that it can be written as

$$g(u) = b_0(1 - 3u^2 + 2u^3) + b_1(3u^2 - 2u^3) + b_2(u - 2u^2 + u^3) + b_3(u^3 - u^2),$$

where

$$\begin{aligned} b_0 &= f(0), \\ b_1 &= f(1), \\ b_2 &= f'(0), \\ b_3 &= f'(1). \end{aligned}$$

That is

$$g(u) = f(0)(1-3u^2+2u^3)+f(1)(3u^2-2u^3)+f'(0)(u-2u^2+u^3)+f'(1)(u^3-u^2).$$

Assume we have the unique cubic polynomial defined by the endpoint values and derivatives as

$$g(u) = a_0 + a_1 u + a_2 u^2 + a_3 u^3,$$

where

$$\begin{aligned} g(0) &= f(0), \\ g(1) &= f(1), \\ g'(0) &= f'(0), \\ g'(1) &= f'(1). \end{aligned}$$

Then we compute

$$b_0 = f(0) = g(0) = a_0,$$

$$b_1 = f(1) = g(1) = a_0 + a_1 + a_2 + a_3.$$

We have

$$g'(u) = a_1 + 2a_2 u + 3a_3 u^2.$$

Hence we compute

$$b_2 = f'(0) = g'(0) = a_1,$$
$$b_3 = f'(1) = g'(1) = a_1 + 2a_2 + 3a_3.$$

Now we solve for the $a$ coefficients in terms of the $b$ coefficients. We get four equations which we call $e1, e2, e3, e4$. We must solve these four equations for the variables $a_0, a_1, a_2, a_3$. We must solve this 4 by 4 linear system symbolically. This is where Maple becomes handy. We use

```
> solve({e1,e2,e3,e4},{a0,a1,a2,a3});
```

Then we substitute the solutions back into

$$g(u) = a_0 + a_1 u + a_2 u^2 + a_3 u^3,$$

getting a cubic polynomial with $b$ coefficients. We collect the coefficients of the powers of $u$ and obtain

$$g = b_0\,(1 - 3\,u^2 + 2\,u^3) + b_1\,(3\,u^2 - 2\,u^3) + b_2\,(u - 2\,u^2 + u^3) \\ + b_3\,(-u^2 + u^3)$$

Here is the maple code to do this:

```
read(hermite);

#compute the basis functions for the hermite interpolant
> g:=a0+a1*u+a2*u^2 +a3*u^3;
```

$$g := a0 + a1\,u + a2\,u^2 + a3\,u^3$$

```
> 'f(u)'=g;
```

$$\mathrm{f}(\,u\,) = a0 + a1\,u + a2\,u^2 + a3\,u^3$$

```
> dg:= diff(g,u);
```

$$dg := a1 + 2\,a2\,u + 3\,a3\,u^2$$

```
> Diff('f(u)',u) = dg;
```

$$\frac{\partial}{\partial u}\,\mathrm{f}(\,u\,) = \mathit{a1} + 2\,\mathit{a2}\,u + 3\,\mathit{a3}\,u^2$$

```
> b0 = 'f(0)';
```

$$\mathit{b0} = \mathrm{f}(\,0\,)$$

```
> b1='f(1)';
```

$$\mathit{b1} = \mathrm{f}(\,1\,)$$

```
> b2=Diff('f(0)',u);
```

$$\mathit{b2} = \frac{\partial}{\partial u}\,\mathrm{f}(\,0\,)$$

```
> b3=Diff('f(1)',u);
```

$$\mathit{b3} = \frac{\partial}{\partial u}\,\mathrm{f}(\,1\,)$$

```
> b0 = subs({u=0},g);e1:=":
```

$$\mathit{b0} = \mathit{a0}$$

```
> b1= subs({u=1},g);e2:=":
```

$$\mathit{b1} = \mathit{a0} + \mathit{a1} + \mathit{a2} + \mathit{a3}$$

```
> b2= subs({u=0},dg);e3:=":
```

$$b2 = a1$$

```
> b3= subs({u=1},dg);e4:=":
```

$$b3 = a1 + 2\,a2 + 3\,a3$$

```
> solve({e1,e2,e3,e4},{a0,a1,a2,a3});
```

$$\{a0 = b0, a1 = b2, a3 = -2\,b1 + 2\,b0 + b2 + b3,$$
$$a2 = -b3 - 2\,b2 + 3\,b1 - 3\,b0\}$$

```
> assign(");
> h:=expand(g);
```

$$h := b0 + b2\,u - u^2\,b3 - 2\,u^2\,b2 + 3\,u^2\,b1 - 3\,u^2\,b0 - 2\,u^3\,b1 + 2\,u^3\,b0$$
$$+ u^3\,b2 + u^3\,b3$$

```
> 'g'=b0*coeff(h,b0)+b1*coeff(h,b1)+b2*coeff(h,b2)+b3*coeff(h,b3);
```

$$g = b0\,(1 - 3\,u^2 + 2\,u^3) + b1\,(3\,u^2 - 2\,u^3) + b2\,(u - 2\,u^2 + u^3)$$
$$+ b3\,(-u^2 + u^3)$$

Here is the Hermite code (hermite.txt) that works in Maple 12:

```
#The Hermite form of a cubic polynomial is specified by
#values and derivative values at 0 and 1.
#Let the polynomial in u be
g:=a0+a1*u+a2*u^2 +a3*u^3;
#The derivative is
dg:= diff(g,u);
Diff('f(u)',u) = dg;
```

```
#define equations where b0,b1,b2,b3 are the end values and derivatives
e1:= b0 = subs({u=0},g);
e2:= b1= subs({u=1},g);
e3:= b2= subs({u=0},dg);
e4:= b3= subs({u=1},dg);
#solve this equations for the coefficients a0,a1,a2,a3
l:=solve({e1,e2,e3,e4},{a0,a1,a2,a3});
# substitute in g
assign(l);
h:=expand(g);
g:=b0*coeff(h,b0)+b1*coeff(h,b1)+b2*coeff(h,b2)+b3*coeff(h,b3);
'g(u)'=g;
```

## 3.6 The Center of a Conic Section

We write The equation of a general conic section as

$$ax^2 + bxy + cy^2 + dx + ey + f = 0.$$

We may write the left side as a quadratic form

$$\psi(P, P) = P^T AP,$$

where $A$ is the symmetric matrix

$$A = \begin{bmatrix} a & b/2 & d/2 \\ b/2 & c & e/2 \\ d/2 & e/2 & f \end{bmatrix}$$

and $P$ is a point:

$$P = \begin{bmatrix} x \\ y \\ 1 \end{bmatrix}.$$

A point $P$ is on the conic section if

$$\psi(P, P) = P^T QP = 0.$$

Given any point $P$, the set

$$\{Q : P^T AQ = 0\}$$

is a straight line, because for $P$ fixed, the equation is linear. This straight line is called the polar of the point $P$. We can show that the polar intersects the conic at tangents to the conic that pass through $P$. See Emery, James D. **Conics, Quadrics, And Projective Space**. Then as $P$ goes to infinity, the polar goes to a diameter. A diameter is a line that is perpendicular to the conic at each intersection point of the line and the conic. The center of a conic is the common intersection point of all diameters. Given an arbitrary conic we shall find the formula for its center. This is useful because the first step in putting a conic into standard form is to translate its center to the origin. Suppose $Q$ is a center, and $P$ is any point at infinity, i.e.

$$P = (p_1, p_2, 0)^T.$$

Then for all such $P$,

$$P^T AQ = 0.$$

Hence a solution is $Q$ were $Q$ satisfies the linear equation

$$AQ = (0, 0, 1)^T.$$

```
% cat min2
# compute the center of a conic section
# see quadric solids report
# conic a x^2 + b xy + cy^2 +d x + e y + f = 0
# or (x,y,1) A (x,y,1)^T = 0
# solve A Q = (0,0,1)^t for the center Q
# use cramers rule homogeneous coordinates
# see conic.ftn, conic.c
with(linalg):
# define quadratic form
A:=array(1..3,1..3):
A[1,1]:=a:
A[1,2]:=b/2:
A[1,3]:=d/2:
A[2,2]:=c:
A[2,3]:=e/2:
A[3,3]:=f:
for i to 3 do for j to 3 do
```

```
  if i>j then A[i,j]:=A[j,i] fi:
od od:
print(A);
dA:=det(A):
Ax:=copy(A):
Ax[1,1]:=0:
Ax[2,1]:=0:
Ax[3,1]:=1:
print(Ax);
dAx:=det(Ax);
Ay:=copy(A);
Ay[1,2]:=0;
Ay[2,2]:=0;
Ay[3,2]:=1;
print(Ay);
dAy:=det(Ay);
Aw:=copy(A):
Aw[1,3]:=0:
Aw[2,3]:=0:
Aw[3,3]:=1:
print(Aw);
dAw:=det(Aw);
x:=dAx/dAw;
y:=dAy/dAw;
readlib(fortran):
fortran(dAx);
fortran(dAy);
fortran(dAw);
```

Here is the Maple session:

```
read(min2);
```

```
# compute the center of a conic section
# see quadric solids report
# conic a x^2 + b xy + cy^2 +d x + e y + f = 0
# or (x,y,1) A (x,y,1)^T = 0
# solve A Q = (0,0,1)^t for the center Q
```

```
# use cramers rule homogeneous coordinates
# see conic.ftn, conic.c
> with(linalg):
Warning: new definition for   norm
Warning: new definition for   trace
# define quadratic form
> A:=array(1..3,1..3):
> A[1,1]:=a:
> A[1,2]:=b/2:
> A[1,3]:=d/2:
> A[2,2]:=c:
> A[2,3]:=e/2:
> A[3,3]:=f:
> for i to 3 do for j to 3 do
>   if i>j then A[i,j]:=A[j,i] fi:
> od od:
> print(A);
```

$$\begin{bmatrix} a & \frac{1}{2}\,b & \frac{1}{2}\,d \\ \frac{1}{2}\,b & c & \frac{1}{2}\,e \\ \frac{1}{2}\,d & \frac{1}{2}\,e & f \end{bmatrix}$$

```
> dA:=det(A):
> Ax:=copy(A):
> Ax[1,1]:=0:
> Ax[2,1]:=0:
> Ax[3,1]:=1:
> print(Ax);
```

$$\begin{bmatrix} 0 & \frac{1}{2}\,b & \frac{1}{2}\,d \\ 0 & c & \frac{1}{2}\,e \\ 1 & \frac{1}{2}\,e & f \end{bmatrix}$$

```
> dAx:=det(Ax);
```

$$dAx := \frac{1}{4}\,b\,e - \frac{1}{2}\,d\,c$$

```
> Ay:=copy(A);
```

$$Ay := \begin{bmatrix} a & \frac{1}{2}\,b & \frac{1}{2}\,d \\ \frac{1}{2}\,b & c & \frac{1}{2}\,e \\ \frac{1}{2}\,d & \frac{1}{2}\,e & f \end{bmatrix}$$

```
> Ay[1,2]:=0;
```

$$Ay_{1,2} := 0$$

```
> Ay[2,2]:=0;
```

$$Ay_{2,2} := 0$$

```
> Ay[3,2]:=1;
```

$$Ay_{3,2} := 1$$

```
> print(Ay);
```

$$\begin{bmatrix} a & 0 & \frac{1}{2}\,d \\ \frac{1}{2}\,b & 0 & \frac{1}{2}\,e \\ \frac{1}{2}\,d & 1 & f \end{bmatrix}$$

```
> dAy:=det(Ay);
```

$$dAy := -\frac{1}{2}\,a\,e + \frac{1}{4}\,b\,d$$

```
> Aw:=copy(A):
> Aw[1,3]:=0:
> Aw[2,3]:=0:
> Aw[3,3]:=1:
> print(Aw);
```

$$\begin{bmatrix} a & \frac{1}{2}\,b & 0 \\ \frac{1}{2}\,b & c & 0 \\ \frac{1}{2}\,d & \frac{1}{2}\,e & 1 \end{bmatrix}$$

```
> dAw:=det(Aw);
```

$$dAw := a\,c - \frac{1}{4}\,b^2$$

```
> x:=dAx/dAw;
```

$$x := \frac{\frac{1}{4}b\ \ e - \frac{1}{2}d\,c}{a\,c - \frac{1}{4}b^2}$$

```
> y:=dAy/dAw;
```

$$y := \frac{-\frac{1}{2}\ \ a\,e + \frac{1}{4}b\,d}{a\,c - \frac{1}{4}b^2}$$

```
> readlib(fortran):
> fortran(dAx);
      t0 = b*e/4-d*c/2
> fortran(dAy);
      t0 = -a*e/2+b*d/4
> fortran(dAw);
      t0 = a*c-b**2/4
```

## 3.7 Solving a Differential Equation With The Laplace Transform

We read the following file into Maple:

```
% cat mlaplace
de:=diff(y(x),x,x)+2*diff(y(x),x)+y(x) = sin(2*x);
dsolve({de,y(0)=1,D(y)(0)=1},y(x));
laplace(de,x,s);
subs(laplace(y(x),x,s)=G,");
solve(",G);
subs({D(y)(0)=1,y(0)=1},");
invlaplace(",s,x);
```

The session is as follows:

```
interface(echo=2);

read(mlaplace);

> de:=diff(y(x),x,x)+2*diff(y(x),x)+y(x) = sin(2*x);
```

$$de := \left(\frac{\partial^2}{\partial x^2}\,\mathrm{y}(\,x\,)\right) + 2\,\left(\frac{\partial}{\partial x}\,\mathrm{y}(\,x\,)\right) + \mathrm{y}(\,x\,) = \sin(\,2\,x\,)$$

```
> dsolve({de,y(0)=1,D(y)(0)=1},y(x));
```

$$\mathrm{y}(\,x\,) = -\frac{4}{25}\cos(\,2\,x\,) - \frac{3}{25}\sin(\,2\,x\,) + \frac{29}{25}\,\mathrm{e}^{(\,-x\,)} + \frac{12}{5}\,\mathrm{e}^{(\,-x\,)}\,x$$

```
> laplace(de,x,s);
```

$$(\,\mathrm{laplace}(\,\mathrm{y}(\,x\,), x, s\,)\,s - \mathrm{y}(\,0\,)\,)\,s - \mathrm{D}(\,y\,)(\,0\,) + 2\,\mathrm{laplace}(\,\mathrm{y}(\,x\,), x, s\,)\,s - 2\,\mathrm{y}(\,0\,) + \mathrm{laplace}(\,\mathrm{y}(\,x\,), x, s\,) = 2\,\frac{1}{s^2+4}$$

```
> subs(laplace(y(x),x,s)=G,");
```

$$(\,G\,s - \mathrm{y}(\,0\,)\,)\,s - \mathrm{D}(\,y\,)(\,0\,) + 2\,G\,s - 2\,\mathrm{y}(\,0\,) + G = 2\,\frac{1}{s^2+4}$$

```
> solve(",G);
```

$$-\frac{-s\,\mathrm{y}(\,0\,) - \mathrm{D}(\,y\,)(\,0\,) - 2\,\mathrm{y}(\,0\,) - 2\,\frac{1}{s^2+4}}{s^2+2\,s+1}$$

```
> subs({D(y)(0)=1,y(0)=1},");
```

$$-\frac{-s-3-2\,\frac{1}{s^2+4}}{s^2+2\,s+1}$$

```
> invlaplace(",s,x);
```

$$-\frac{4}{25}\cos(\,2\,x\,)-\frac{3}{25}\sin(\,2\,x\,)+\frac{29}{25}\,\mathrm{e}^{(\,-x\,)}+\frac{12}{5}\,\mathrm{e}^{(\,-x\,)}\,x$$

The solution using dsolve, and the solution using the Laplace transform method are the same.

## 3.8 Matrices: A Least Squares Problem

We will compute a least squares solution of

$$c_1x^2+c_2x+c_3=y$$

Given a set of $n$ data points, we may write the $n$ equations as

$$ac=b,$$

which is an overdetermined linear system if $n>3$. $a$ is a $n$ by 3 matrix defined by the $n$ $x$ values, $c$ is the 3 dimensional coefficient vector, and $b$ is a $n$ dimensional vector of $y$ values. Then the normal equation is the three by three linear system

$$a^tac=a^tb.$$

Here is the file that we will read:

```
% cat min3
# least squares solution of c1 x^2 + c2 x + c3 = y
with(linalg);
# define data ( 5 points)
x:=array(1..5,1..1,[
```

```
[1],
[2],
[3],
[4],
[5]
]);
y:=array(1..5,1..1,[
[1],
[4],
[10],
[16],
[24]
]);
a:=array(1..5,1..3);
for i to 5 do for j to 3 do
  if j=1 then a[i,j]:=1 fi;
  if j=2 then a[i,j]:=x[i,1] fi;
  if j=3 then a[i,j]:=x[i,1]^2 fi;
od od;
print(a);
at:=transpose(a);
ata:=evalm(at &* a);
aty:=evalm(at &* y);
atainv:=inverse(ata);
c:=evalm(atainv &* aty);
Digits:=40;
evalf(c[1,1]);
evalf(c[2,1]);
evalf(c[3,1]);
```

Here is the interactive session:

```
interface(echo=2);

read(min3);

# least squares solution of c1 x^2 + c2 x + c3 = y
```

```
> with(linalg);
Warning: new definition for   norm
Warning: new definition for   trace


# define data ( 5 points)
> x:=array(1..5,1..1,[
> [1],
> [2],
> [3],
> [4],
> [5]
> ]);
```

$$x := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \end{bmatrix}$$

```
> y:=array(1..5,1..1,[
> [1],
> [4],
> [10],
> [16],
> [24]
> ]);
```

$$y := \begin{bmatrix} 1 \\ 4 \\ 10 \\ 16 \\ 24 \end{bmatrix}$$

```
> a:=array(1..5,1..3);
```

$$a := \mathrm{array}(\,1..5, 1..3, [\,]\,)$$

```
> for i to 5 do for j to 3 do
>   if j=1 then a[i,j]:=1 fi;
>   if j=2 then a[i,j]:=x[i,1] fi;
>   if j=3 then a[i,j]:=x[i,1]^2 fi;
> od od;
> print(a);
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \\ 1 & 4 & 16 \\ 1 & 5 & 25 \end{bmatrix}$$

```
> at:=transpose(a);
```

$$at := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 3 & 4 & 5 \\ 1 & 4 & 9 & 16 & 25 \end{bmatrix}$$

```
> ata:=evalm(at &* a);
```

$$ata := \begin{bmatrix} 5 & 15 & 55 \\ 15 & 55 & 225 \\ 55 & 225 & 979 \end{bmatrix}$$

```
> aty:=evalm(at &* y);
```

$$aty := \begin{bmatrix} 55 \\ 223 \\ 963 \end{bmatrix}$$

```
> atainv:=inverse(ata);
```

$$atainv := \begin{bmatrix} \frac{23}{5} & \frac{-33}{10} & \frac{1}{2} \\ \frac{-33}{10} & \frac{187}{70} & \frac{-3}{7} \\ \frac{1}{2} & \frac{-3}{7} & \frac{1}{14} \end{bmatrix}$$

```
> c:=evalm(atainv &* aty);
```

$$c := \begin{bmatrix} \frac{-7}{5} \\ \frac{53}{35} \\ \frac{5}{7} \end{bmatrix}$$

```
> Digits:=40;
```

$$Digits := 40$$

```
> evalf(c[1,1]);
```

$$-1.400000000000000000000000000000000000000$$

```
> evalf(c[2,1]);
```

$$1.514285714285714285714285714285714285714$$

```
> evalf(c[3,1]);
```

$$.7142857142857142857142857142857142857143$$

## 3.9 Show Curl and Divergence Of A Point Coulomb Source Is Zero.

We will show that if vector field $f$ varies as the inverse of the distance squared from a point, and is directed away from the point, then

$$\nabla \times f = 0,$$

and

$$\nabla \cdot f = 0.$$

The file is min4.

```
% cat min4
# compute curl and divergence of point coulomb source
# show they are zero
r:=sqrt(x^2+y^2+z^2);
v:=[x,y,z];
f:=[x/r^3,y/r^3,z/r^3];
with(linalg);
curl(f,v);
# show also holds for translated coordinates
g:=subs(x=x-xp,y=y-yp,z=z-zp,f);
a:=curl(g,v);
a[1]:=normal(a[1]);
a[2]:=normal(a[2]);
a[3]:=normal(a[3]);
print(a);
b:=diverge(f,v);
normal(b);
```

```
interface(echo=2);
```

```
read(min4);
```

```
# compute curl and divergence of point coulomb source
# show they are zero
> r:=sqrt(x^2+y^2+z^2);
```

$$r := \sqrt{x^2+y^2+z^2}$$

```
> v:=[x,y,z];
```

$$v := [x, y, z]$$

```
> f:=[x/r^3,y/r^3,z/r^3];
```

$$f := \left[\frac{x}{(x^2+y^2+z^2)^{3/2}}, \frac{y}{(x^2+y^2+z^2)^{3/2}}, \frac{z}{(x^2+y^2+z^2)^{3/2}}\right]$$

```
> with(linalg);
Warning: new definition for   norm
Warning: new definition for   trace
```

[*BlockDiagonal*, *GramSchmidt*, *JordanBlock*, *Wronskian*, *add*, *addcol*, *addrow*, *adj*, *adjoint*, *angle*, *augment*, *backsub*, *band*, *basis*, *bezout*, *blockmatrix*, *charmat*, *charpoly*, *col*, *coldim*, *colspace*, *colspan*, *companion*, *concat*, *cond*, *copyinto*, *crossprod*, *curl*, *definite*, *delcols*, *delrows*, *det*, *diag*, *diverge*, *dotprod*, *eigenvals*, *eigenvects*, *entermatrix*, *equal*, *exponential*, *extend*, *ffgausselim*, *fibonacci*, *frobenius*, *gausselim*, *gaussjord*, *genmatrix*, *grad*, *hadamard*, *hermite*, *hessian*, *hilbert*, *htranspose*, *ihermite*, *indexfunc*, *innerprod*, *intbasis*, *inverse*, *ismith*, *iszero*, *jacobian*, *jordan*, *kernel*, *laplacian*, *leastsqrs*, *linsolve*, *matrix*, *minor*, *minpoly*, *mulcol*, *mulrow*, *multiply*, *norm*, *normalize*, *nullspace*, *orthog*, *permanent*, *pivot*, *potential*, *randmatrix*, *randvector*, *rank*, *ratform*, *row*, *rowdim*, *rowspace*, *rowspan*, *rref*, *scalarmul*, *singularvals*, *smith*, *stack*, *submatrix*, *subvector*, *sumbasis*, *swapcol*, *swaprow*, *sylvester*, *toeplitz*, *trace*, *transpose*, *vandermonde*, *vecpotent*, *vectdim*, *vector*]

```
> curl(f,v);
```

$$[0\,0\,0]$$

```
# show also holds for translated coordinates
> g:=subs(x=x-xp,y=y-yp,z=z-zp,f);
```

$$g := \left[\frac{x - xp}{((x - xp)^2 + (y - yp)^2 + (z - zp)^2)^{3/2}},\right.$$
$$\frac{y - yp}{((x - xp)^2 + (y - yp)^2 + (z - zp)^2)^{3/2}},$$
$$\left.\frac{z - zp}{((x - xp)^2 + (y - yp)^2 + (z - zp)^2)^{3/2}}\right]$$

```
> a:=curl(g,v);
```

$$a := \left[-\frac{3}{2}\frac{(z - zp)(2y - 2yp)}{\%1^{5/2}} + \frac{3}{2}\frac{(y - yp)(2z - 2zp)}{\%1^{5/2}}\right.$$
$$-\frac{3}{2}\frac{(x - xp)(2z - 2zp)}{\%1^{5/2}} + \frac{3}{2}\frac{(z - zp)(2x - 2xp)}{\%1^{5/2}}$$
$$\left.-\frac{3}{2}\frac{(y - yp)(2x - 2xp)}{\%1^{5/2}} + \frac{3}{2}\frac{(x - xp)(2y - 2yp)}{\%1^{5/2}}\right]$$
$$\%1 := (x - xp)^2 + (y - yp)^2 + (z - zp)^2$$

```
> a[1]:=normal(a[1]);
```

$$a_1 := 0$$

```
> a[2]:=normal(a[2]);
```

$$a_2 := 0$$

```
> a[3]:=normal(a[3]);
```

$$a_3 := 0$$

```
> print(a);
```

$$[0\,0\,0]$$

```
> b:=diverge(f,v);
```

$$b := 3\,\frac{1}{(x^2+y^2+z^2)^{3/2}} - 3\,\frac{x^2}{(x^2+y^2+z^2)^{5/2}} - 3\,\frac{y^2}{(x^2+y^2+z^2)^{5/2}} - 3\,\frac{z^2}{(x^2+y^2+z^2)^{5/2}}$$

```
> normal(b);
```

$$0$$

## 3.10 Potential of an Axial Quadrupole

```
#3.8 Problem axial quadrupole, charges q, -2q, q separated by s.
# distances from charges are a,r and b.
#p:=q/(4*pi*e0*r)*(r/a+r/b-2);
#Law of cosines:
#a:=sqrt(r^2+s^2-2*r*s*cos(t));
#b:=sqrt(r^2+s^2+2*r*s*cos(t));
#u=(s/r)^2,
```

```
#v=2*s*cos(t)/r
f:=(1+w)^(-1/2);
g:=taylor(f,w=0,4);
h:=convert(g,polynom);
p:=subs(w=u-v,h)+subs(w=u+v,h)-2;
p:=expand(p);
p:=subs(u=(s/r)^2,v=2*s*cos(t)/r,p);
# keep only terms in s^2:
p:=s*s*coeff(p,s,2);
p:=expand(p);
p:=p*q/(4*pi*e0*r);
normal(p);
```

## 3.11 Helix, Resultant

```
% cat min10
cx:=(r+h+u^2*(h-r))/(1+u*u);
cx:=subs(h=0,k=0,cx);
cy:= (2*r*u+k+k*u^2)/(1+u*u);
cy:=subs(h=0,k=0,cy);
h1x:=subs(r=r1,cx);
h1y:=subs(r=r1,cy);
h1z:=b1+c*u;
h2x:=subs(r1=r2,h1x);
h2y:=subs(r1=r2,h1y);
h2z:=subs(b1=0,h1z);
hx:=(1-v)*h1x+v*h2x;
hx:=normal(hx);
hy:=(1-v)*h1y+v*h2y;
hy:=normal(hy);
hz:=(1-v)*h1z+v*h2z;
e1:=x*denom(hx)-numer(hx);
vv:=solve(hz=z,v);
e1:=subs(v=vv,e1);
e1:=normal(e1);
e1:=numer(e1);
```

```
e2:=y*denom(hy)-numer(hy);
e2:=subs(v=vv,e2);
e2:=normal(e2);
e2:=numer(e2);
e3:=resultant(e1,e2,u);
with(linalg);
cf:=array(0..3,0..3,0..4);
k:=0;
for i from 0 to 3 do
for j from 0 to 3 do
for k from 0 to 4 do
cf[i,j,k]:=coeff(coeff(coeff(e3,x,i),y,j),z,k);
if (cf[i,j,k] <> 0) then print(i,j,k) fi;
od
od
od;
e4:=0;
for i from 0 to 3 do
for j from 0 to 3 do
for k from 0 to 4 do
if (cf[i,j,k] <> 0) then e4:=e4+cf[i,j,k]*x^(i)*y^(j)*z^(k) fi;
od
od
od;
e4;
e5:=normal(e4-e3);
```

## 3.12 Helix

```
% cat min11
cx:=(r+h+u^2*(h-r))/(1+u*u);
cx:=subs(h=0,k=0,r=1,cx);
cy:= (2*r*u+k+k*u^2)/(1+u*u);
cy:=subs(h=0,k=0,r=1,cy);
hx:=cx*v;
hy:=cy*v;
```

```
hz:=c2*u+(c1+c2*u)*v;
hx:=normal(hx);
hy:=normal(hy);
e1:=x*denom(hx)-numer(hx);
vv:=solve(hz=z,v);
e1:=subs(v=vv,e1);
e1:=normal(e1);
e1:=numer(e1);
e2:=y*denom(hy)-numer(hy);
e2:=subs(v=vv,e2);
e2:=normal(e2);
e2:=numer(e2);
e3:=resultant(e1,e2,u);
e3:=e3/(4*c2);
with(linalg);
cf:=array(0..6,0..6,0..6);
for i from 0 to 6 do
for j from 0 to 6 do
for k from 0 to 6 do
cf[i,j,k]:=coeff(coeff(coeff(e3,x,i),y,j),z,k);
if (cf[i,j,k] <> 0) then print(i,j,k) fi;
od
od
od;
e4:=0;
for i from 0 to 6 do
for j from 0 to 6 do
for k from 0 to 6 do
if (cf[i,j,k] <> 0) then e4:=e4+cf[i,j,k]*x^(i)*y^(j)*z^(k) fi;
od
od
od;
e4;
e5:=normal(e4-e3);
readlib(fortran);
fortran(e3);
e6:=subs(z=0,e3);
```

```
fortran(e6);
e7:=subs(z=0,c1=1,c2=1,e3);
fortran(e7);
```

## 3.13 Helicoid, Parametric and Implicit Representation

```
% cat min12
# define helicoid as a rational parametric surface
# and find implicit representation
# define rational parametric circle
cx:=(r+h+u^2*(h-r))/(1+u*u);
cx:=subs(h=0,k=0,r=1,cx);
cy:= (2*r*u+k+k*u^2)/(1+u*u);
cy:=subs(h=0,k=0,r=1,cy);
# defind curve moving up z-axis
r1x:=0;
r1y:=0;
r1z:=c2*u;
# define rational parametric spiral on
# unit cylinder
r2x:=cx;
r2y:=cy;
r2z:=c1+c2*u;
# define helicoid as ruled surface
hx:=r1x+v*(r2x-r1x);
hy:=r1y+v*(r2y-r1y);
hz:=r1z+v*(r2z-r1z);
hx:=normal(hx);
hy:=normal(hy);
# convert equation x=hx into a polynomial
e1:=x*denom(hx)-numer(hx);
# solve for v and substitute in first equation
vv:=solve(hz=z,v);
e1:=subs(v=vv,e1);
```

```
e1:=normal(e1);
e1:=numer(e1);
# convert equation y=hy into a polynomial
e2:=y*denom(hy)-numer(hy);
# substitute for v in second equation
e2:=subs(v=vv,e2);
e2:=normal(e2);
e2:=numer(e2);
# eliminate u from first and second equation
e3:=resultant(e1,e2,u);
# load routine for generating fortran expression;
readlib(fortran);
fortran(e3);
e4:=factor(e3);
fortran(e4);
# an extraneous factor has been introduced
# check that a point on parametric surface is on
# implicit surface
uu:=9;
vv:=12;
xx:=subs(u=uu,v=vv,hx);
yy:=subs(u=uu,v=vv,hy);
zz:=subs(u=uu,v=vv,hz);
e5:=subs(x=xx,y=yy,z=zz,e3);
e5:=expand(e5);
```

## 3.14 Approximate Helicoid

Define an approximate helicoid as a rational parametric surface and find an implicit representation. Define a rational parametric circle.

Let a circle have radius $r$ and center $(r, 0)$. Let the line with slope $t$,

$$y = tx$$

intersect the circle at $(x, y)$. The circle has equation

$$(x - r)^2 + y^2 = r^2.$$

Solving these two equations

$$x = \frac{2r}{1+t^2}$$

$$y = \frac{2tr}{1+t^2}$$

Translating the circle so that it has center $(h, k)$ we find

$$x = \frac{r + h + t^2(h - r)}{1+t^2}$$

and

$$y = \frac{2rt + k + kt^2}{1+t^2}.$$

Now we wish to represent the arc of the circle from angle $-\alpha$ to $\alpha$ where $\alpha$ is less than $\pi$. Thus $t$ ranges between $\tan(-\alpha/2)$ and $\tan(\alpha/2)$. For an arbitrary arc of angle less than $2\pi$ from $\alpha_1$ to $\alpha_2$ we let

$$\phi = \frac{\alpha_1 + \alpha_2}{2}.$$

Let $s = \sin(\phi)$ and $c = \cos(\phi)$ . We rotate by $-\phi$ that is $x$ goes to $cx + sy$ and $y$ goes to $-sx + cy$. Let $t_0 = \tan(|\alpha_2 - \alpha_1|/4)$ Then $t$ ranges from $-t_0$ to $t_0$. The arc is a rational quadratic curve.

A helicoid is a ruled surface generated by a line, where the points on the line move in helical paths. Let

$$R_1(w) = c_2 wk.$$

and

$$R_2(w) = \cos(w)i + \sin(w)j + (c_1 + c_2 w)k$$

$R_2$ is a helix. The helicoid is

$$s(w, v) = R_1(w) + v(R_2(w) - R_1(w)).$$

If one thinks of this as a screw thread, then $c_1$ defines the thread angle (i.e. 30 degrees) and $c_2$ defnes the screw pitch. Let us use the rational representation of the unit circle. We get

$$\cos(w) = \frac{1-u^2}{1+u^2}$$

and

$$\sin(w) = \frac{2u}{1+u^2},$$

where

$$u = \tan(w/2).$$

Approximate $\tan(w/2)$ by $w/2$. Then $w = 2u$. We get a rational approximation to the helicoid $S(u,v)$. This is a rational surface of order three. Eliminate $u$ and $v$. Use resultant. (see books on algebraic curves and algebraic geometry for explanation of resultant).
Implicit equation in $x$, $y$, and $z$ calculated with Maple. (Maple is a computer algebra system from the University of Waterloo)
If we use the rational representation of the circle, we can express $\mathbf{R}_1$ as a rational function of say $u$. We have

$$x(u,v) = \frac{1-u^2}{1+u^2}v,$$

$$y(u,v) = \frac{2u}{1+u^2}v,$$

and

$$z(u,v) = c_2u + (c_1 + c_2u)v.$$

This is a rational surface of order three. We may eliminate $u$ and $v$ to get an implicit representation. We solve the third equation for $v$ and substitute in the first and second equations. Then we calculate the resultant to eliminate $u$ from the remaining two equations. We get a single implicit equation in $x$, $y$ and $z$.

We get the implicit equation

$$4c_1(z^2 + 4c_2^2)(4yc_2^2 + 4c_2c_1y^2 + 4c_2c_1x^2$$

$$-4zxc_2 + c_1^2x^2y + c_1^2y^3 - yz^2).$$

Use of the resultant has introduced an extraneous factor

$$(z^2 + 4c_2^2).$$

The implicit cubic helicoid surface is

$$4yc_2^2 + 4c_2c_1y^2 + 4c_2c_1x^2$$

$$-4zxc_2 + c_1^2x^2y + c_1^2y^3 - yz^2.$$

The Maple input file that produced this is:

```
% cat min13
# define approximate helicoid as a rational parametric surface
# and find implicit representation
# define rational parametric circle
cx:=(r+h+u^2*(h-r))/(1+u*u);
cx:=subs(h=0,k=0,r=1,cx);
cy:= (2*r*u+k+k*u^2)/(1+u*u);
cy:=subs(h=0,k=0,r=1,cy);
# define curve moving up z-axis
r1x:=0;
r1y:=0;
r1z:=2*c2*u;
# define rational parametric curve on
# unit cylinder
r2x:=cx;
r2y:=cy;
r2z:=c1+2*c2*u;
# define helicoid as ruled surface
hx:=r1x+v*(r2x-r1x);
hy:=r1y+v*(r2y-r1y);
hz:=r1z+v*(r2z-r1z);
hx:=normal(hx);
hy:=normal(hy);
# convert equation x=hx into a polynomial
e1:=x*denom(hx)-numer(hx);
# solve for v and substitute in first equation
vv:=solve(hz=z,v);
e1:=subs(v=vv,e1);
```

```
e1:=normal(e1);
e1:=numer(e1);
# convert equation y=hy into a polynomial
e2:=y*denom(hy)-numer(hy);
# substitute for v in second equation
e2:=subs(v=vv,e2);
e2:=normal(e2);
e2:=numer(e2);
# eliminate u from first and second equation
e3:=resultant(e1,e2,u);
# load routine for generating fortran expression;
readlib(fortran);
fortran(e3);
e4:=factor(e3);
fortran(e4);
# an extraneous factor has been introduced
# check that a point on parametric surface is on
# implicit surface
uu:=9;
vv:=12;
xx:=subs(u=uu,v=vv,hx);
yy:=subs(u=uu,v=vv,hy);
zz:=subs(u=uu,v=vv,hz);
e5:=subs(x=xx,y=yy,z=zz,e3);
e5:=expand(e5);
% cat min14
a:=(v1-v3)/(exp(b*u1)-exp(b*u3));
c:=v1-a*exp(b*u1);
f:=a*exp(b*u2)+c-v2;
```

## 3.15 Discriminant 1

```
% cat min15
f:=a*x^3+b*x^2+c*x+d;
df:=diff(f,x);
ddf:=diff(df,x);
```

```
xi:=solve(ddf=0,x);
g:=expand(subs(x=z+xi,f));
h:=a*z^3+c1*z+d1;
dh:=diff(h,z);
ddh:=diff(dh,z);
dddh:=diff(ddh,z);
q:=dddh*(1+dh*dh)-3*ddh^2*dh;
q:=expand(q);
A:=coeff(q,z,4);
B:=coeff(q,z,2);
C:=coeff(q,z,0);
q1:=A*w*w+B*w+C;
discriminant:=factor(B*B-4*A*C);
x1:= xi + (w1)^(1/2);
x2:=-xi - (w1)^(1/2);
w1:=radsimp((-B-sqrt(B*B-4*A*C))/(2*A));
w2:=radsimp((-B+sqrt(B*B-4*A*C))/(2*A));
c1=coeff(g,z,1);
```

## 3.16 Discriminant 2

```
% cat min15
f:=a*x^3+b*x^2+c*x+d;
df:=diff(f,x);
ddf:=diff(df,x);
xi:=solve(ddf=0,x);
g:=expand(subs(x=z+xi,f));
h:=a*z^3+c1*z+d1;
dh:=diff(h,z);
ddh:=diff(dh,z);
dddh:=diff(ddh,z);
q:=dddh*(1+dh*dh)-3*ddh^2*dh;
q:=expand(q);
A:=coeff(q,z,4);
B:=coeff(q,z,2);
C:=coeff(q,z,0);
```

```
q1:=A*w*w+B*w+C;
discriminant:=factor(B*B-4*A*C);
x1:= xi + (w1)^(1/2);
x2:=-xi - (w1)^(1/2);
w1:=radsimp((-B-sqrt(B*B-4*A*C))/(2*A));
w2:=radsimp((-B+sqrt(B*B-4*A*C))/(2*A));
c1=coeff(g,z,1);
% cat min16
t:=arccos(abs(s)/a)-arccos((a^2+d^2-r^2)/(2*a*d));
t:=subs(a=(s^2+r^2)^(1/2),t);
dtds:=diff(t,s);
dtds:=subs(s^2+r^2=a^2,dtds);
```

## 3.17 Arccos

```
% cat min16
t:=arccos(abs(s)/a)-arccos((a^2+d^2-r^2)/(2*a*d));
t:=subs(a=(s^2+r^2)^(1/2),t);
dtds:=diff(t,s);
dtds:=subs(s^2+r^2=a^2,dtds);
```

## 3.18 Matrix, Transpose

```
% cat min17
f:=arccos(s/(s*s+r*r)^(1/2))-arccos((s*s+d*d)/(2*d*(s*s+r*r)^(1/2)));
% cat min18
with(linalg);
x:=array(1..5,1..1,[[x1],[x2],[x3],[x4],[x5]]);
y:=array(1..5,1..1,[[y1],[y2],[y3],[y4],[y5]]);
B:=array(1..5,1..1,[[x1^2],[x2^2],[x3^2],[x4^2],[x5^2]]);
A:=array(1..5,1..2,[[y1^2,y1],[y2^2,y2],[y3^2,y3],[y4^2,y4],[y5^2,y5]]);
c:=array(1..2,1..1,[[c1],[c2]]);
evalm(A &* c)=B;
At:=transpose(A);
evalm((At &* A) &* c) = evalm(At &* B);
```

```
% cat min18
```

## 3.19 Matrix, Transpose

```
with(linalg);
x:=array(1..5,1..1,[[x1],[x2],[x3],[x4],[x5]]);
y:=array(1..5,1..1,[[y1],[y2],[y3],[y4],[y5]]);
B:=array(1..5,1..1,[[x1^2],[x2^2],[x3^2],[x4^2],[x5^2]]);
A:=array(1..5,1..2,[[y1^2,y1],[y2^2,y2],[y3^2,y3],[y4^2,y4],[y5^2,y5]]);
c:=array(1..2,1..1,[[c1],[c2]]);
evalm(A &* c)=B;
At:=transpose(A);
evalm((At &* A) &* c) = evalm(At &* B);
```

## 3.20 Equation Solving

```
interface(echo=2);

read(min19);

> e1:=(a12+a123+a13)/(a12+a123+a13+a2+a23+a3)=4/10;
```

$$e1 := \frac{a12 + a123 + a13}{a12 + a123 + a13 + a2 + a23 + a3} = \frac{2}{5}$$

```
> e2:=(a12+a123+a23)/(a12+a123+a23+a1+a13+a3)=3/10;
```

$$e2 := \frac{a12 + a123 + a23}{a12 + a123 + a23 + a1 + a13 + a3} = \frac{3}{10}$$

```
> e3:=(a23+a123+a13)/(a23+a123+a13+a2+a12+a1)=2/10;
```

$$e3 := \frac{a23 + a123 + a13}{a23 + a123 + a13 + a2 + a12 + a1} = \frac{1}{5}$$

```
> e4:=a123/(a123+a23)=9/10;
```

$$e4 := \frac{a123}{a123 + a23} = \frac{9}{10}$$

```
> e5:=a123/(a123+a13)=8/10;
```

$$e5 := \frac{a123}{a123 + a13} = \frac{4}{5}$$

```
> e6:=a123/(a123+a12)=7/10;
```

$$e6 := \frac{a123}{a123 + a12} = \frac{7}{10}$$

```
> e7:=a1+a12+a123+a13=1/100;
```

$$e7 := a1 + a12 + a123 + a13 = \frac{1}{100}$$

```
> solve({e1,e2,e3,e4,e5,e6,e7},{a1,a2,a3,a12,a13,a23,a123});
```

$$\left\{a12 = \frac{324}{351875}, a3 = \frac{1109}{1407500}, a123 = \frac{756}{351875}, a2 = \frac{6169}{1407500},\right.$$
$$\left. a1 = \frac{8999}{1407500}, a13 = \frac{189}{351875}, a23 = \frac{84}{351875}\right\}$$

```
> v:=evalf(");
```

$$\begin{aligned} v := \{&a123 = .002148490231, a3 = .0007879218472,\\ &a12 = .0009207815275, a2 = .004382948490,\\ &a1 = .006393605684, a13 = .0005371225577,\\ &a23 = .0002387211368\}\end{aligned}$$

```
> b1:=8999/1407500;
```

$$b1 := \frac{8999}{1407500}$$

```
> b2:=6169/1407500;
```

$$b2 := \frac{6169}{1407500}$$

```
> b3=1109/1407500;
```

$$b3 = \frac{1109}{1407500}$$

```
> b12:=324/351875;
```

$$b12 := \frac{324}{351875}$$

```
> b13:=189/351875;
```

$$b13 := \frac{189}{351875}$$

```
> b23:=84/351875;
```

$$b23 := \frac{84}{351875}$$

```
> b123:=756/351875;
```

$$b123 := \frac{756}{351875}$$

```
> b2+b12+b123+b13+b3;
```

$$\frac{2249}{281500} + b3$$

```
> c2:=";
```

$$c2 := \frac{2249}{281500} + b3$$

```
> b3:=1109/1407500;
```

$$b3 := \frac{1109}{1407500}$$

```
> c2;
```

$$\frac{6177}{703750}$$

```
> c1:=b1+b12+b123+b13;
```

$$c1 := \frac{1}{100}$$

```
> c3:=a3+a13+a123+a23;
```

$$c3 := a3 + a13 + a123 + a23$$

```
> c3:=b3+b13+b123+b23;
```

$$c3 := \frac{209}{56300}$$

## 3.21 The Inverse of a Bilinear Transformation

```
% cat min20
# min20,inverse of bilinear transformation geometry.tex 11/29/93
# to run: read(min20);
e1:=(x -x1 -x3*v);
e2:=(y2 - y4*v);
e3:=(y - y1-y3*v);
e4:= (x2 + x4*v);
e5:= e1*e2 - e3*e4;
e6:=expand(e5);
collect(e6,v);
a:=coeff(e6,v,2);
b:=coeff(e6,v,1);
c:=coeff(e6,v,0);
e7:=expand(a*v*v+b*v+c);
e8:=e6-e7;
#       a := x3 y4 + y3 x4
#       b := - x y4 + x1 y4 + y1 x4 - y x4 - x3 y2 + y3 x2
#       c := x y2 - x1 y2 + y1 x2 - y x2
```

## 3.22 Equivalent Circuit Impedance

```
% cat min21
# min21, piezoelectric equivalent impedance 5/5/94
# to run: read(min21);
zm:=r+I*(m*w - k/w);
p:=-I*w*a1+ I*w*a2/(1 - 1/(I*w*zm));
z:=1/p;
z:=evalc(z);
z:=radsimp(z);
zn:=numer(z);
zd:=denom(z);
zni:=coeff(zn,I,1);
zdi:=coeff(zd,I,1);
```

```
c2:=coeff(zni,w,2);
c4:=coeff(zni,w,4);
c0:=coeff(zni,w,0);
rem := expand(zni - (c0 + c2*w^2 + c4*w^4));

% cat min22
# min22, piezoelectric equivalent impedance 5/5/94
# to run: read(min21);
zm:=r+I*(m*w - k/w);
p:=-I*w*a1+ I*w*a2/(1 - 1/(I*w*zm));
z:=1/p;
z:=evalc(z);
z:=radsimp(z);
z:=evalc(z);
zi:=coeff(z,I,1);
solve(zi=0,w);

% cat min23
# min23, piezoelectric equivalent impedance 5/5/94
# to run: read(min21);
zm:=r+I*(m*w - k/w);
p:=-I*w*a1+ I*w*a2/(1 - 1/(I*w*zm));
r:=5;
m:=10;
k:=7;
a1:=13;
a2:=17;
z:=1/p;
z:=evalc(z);
z:=radsimp(z);
z:=evalc(z);
zi:=coeff(z,I,1);
solve(zi=0,w);

% cat min24
# min24, piezoelectric equivalent impedance 5/5/94
# to run: read(min21);
```

```
zm:=r+I*(m*w - k/w);
p:=-I*w*a1 - I*w*a2/(a3 + 1/(I*w*zm));
z:=1/p;
z:=evalc(z);
z:=radsimp(z);
z:=evalc(z);
zi:=coeff(z,I,1);
nzi:=numer(zi);
c4:=coeff(nzi,w,4);
c2:=coeff(nzi,w,2);
c0:=coeff(nzi,w,0);
d:=expand(c2-4*c4*c0);
readlib(fortran);
fortran(c4);
fortran(c2);
fortran(c0);
```

## 3.23 Compute A Jacobian Matrix For A FORTRAN Program

```
% cat minjac
f:=a*x^b*(exp(-c*x)+d)-y;
f1:=subs(x=xm,y=ym,f);
f2:=subs(x=xh,y=yh,f);
f3:=subs(x=xl,y=yl,f);
f4:=(b-c*x)*exp(-c*x)+b*d;
df1da:=diff(f1,a);
df2da:=diff(f2,a);
df3da:=diff(f3,a);
df4da:=diff(f4,a);
df1db:=diff(f1,b);
df2db:=diff(f2,b);
df3db:=diff(f3,b);
df4db:=diff(f4,b);
df1dc:=diff(f1,c);
```

```
df2dc:=diff(f2,c);
df3dc:=diff(f3,c);
df4dc:=diff(f4,c);
df1dd:=diff(f1,d);
df2dd:=diff(f2,d);
df3dd:=diff(f3,d);
df4dd:=diff(f4,d);
readlib(fortran);
fortran(df1da);
fortran(df2da);
fortran(df3da);
fortran(df4da);
fortran(df1db);
fortran(df2db);
fortran(df3db);
fortran(df4db);
fortran(df1dc);
fortran(df2dc);
fortran(df3dc);
fortran(df4dc);
fortran(df1dd);
fortran(df2dd);
fortran(df3dd);
fortran(df4dd);
```

## 3.24 Bernstein Polynomials

```
% cat bernstein
# bernstein polynomial 3/23/92
bernstein := proc(n,k,x)
local v,p;
v := binomial(n,k) * (1-x)^(n-k)*x^k;
if k < 0 then v:= 0 fi;
if k > n then v:= 0 fi;
v;
end;
```

## 3.25 Equivalent Computer Algebra Commands For Various Systems

```
% cat calg.scr
Computer Algebra Systems:
(Maple, Mathematica, Reduce, Derive, Macsyma, SMP, Mumath, Cayley,...))
Computer Algebra Commands
SMP                MAPLE                MUMATH

smp                maple                MUSIMP MATSOL
                                        LOAD(CALCULUS);
                                        BELL: FALSE
%                 "                     @
f:Sin[x]          f:=sin(x);            F:SIN(X);
D[f,x]            diff(f,x);            DIF(F,X);
Int[f,x]          int(f,x);             INT(F,X);
Ps[f,x,0,5]       taylor(f,x=0,5);      TAYLOR(F,X,0,5);
a                 print(a);             A;
A:{{a,b},{c,d}}                         A:{[A,B],[C,D]};
                  A:=array([a,b],[c,d]);

                  linsolve(A,B);        LINEQN([X+Y==1,X-Y==2,[X,Y]);
e[2]              op(2,e);              THIRD(E);
Ex[a]             expand(a);            EXPAND(A);
Fac[e]           factor(e);
cntrl/z           quit                  SYSTEM();
Rat[e]            normal(e);            EXPD(A);DIVOUT(A,X);
Col[e]            collect(e,x);
                  radsimp(e);
Sol[{e1=0,e2=0},{s,t}]                  SECOND(SOLVE(X^2==-1));
                   Solve({e1=0,e2=0},{s,t});

<"test"            read map;            LOAD(CALCULUS);
<xrerun
Rerun["eq.in"]
```

```
Num[e]              numer(e);              NUM(E);
Den[e]              denom(e);              DEN(E);
Pdiv[n,d,c]         quo(c,d,x,'r');        PQUOT(C,D,X);
Coef[x^3,e]         coef(e,x,3);           COEFF(SECOND(REST(E)));

S[e,t->x+Sin[x]]    subs(t=x+sin(x),e);    EVSUB(E,T,X+SIN(X));
                                           EVAL(E);
                                           PARFRAC(E,X);
                                           RATIONALIZE((2+#I)/(1+3 #I));
X:                                         X:'X
                                           POINT: -6;
                                           TRGEXPD(COS(X)^4,3);
```

## 3.26 Elastic Constants

```
# elastic 4/21/92
# elastic constants for an isotropic solid
# s=c e, where s is a vector of the stress components,
# e is a vector of the strain components, and c is the 6 by 6
# elastic matrix. for an isotropic solid this reduces to the
# equations:
# (l+2*m)*e1+l*e2+l*e3 = s1;
# l*e1 + (l+2*m)*e2 + l*e3 = s2;
# l*e1 + l*e2 + (l+2*m)*e3 = s3;
# m*e12 = s12
# m*e13 = s13
# m*e23 = s23
# where l and m are the lame constants.
# suppose a rod is under only tensil stress in one direction
# then the first three equations become:
f1:=(l+2*m)*e1+l*e2+l*e3 = s1;
f2:=l*e1 + (l+2*m)*e2 + l*e3 = 0;
f3:= l*e1 + l*e2 + (l+2*m)*e3 = 0;
solve({f1,f2,f3},{e1,e2,e3});
# solving we find the strains
#                l s1                     l s1            s1 (l + m)
```

```
# {e3 = - 1/2 -------------, e2 = - 1/2 -------------, e1 = -------------}
#              m (3 l + 2 m)              m (3 l + 2 m)       m (3 l + 2 m)
e1:=s1*(l+m)/(m*(3*l+2*m));
e2:=-l*s1/(2*m*(3*l+2*m)) ;
# young's modulus is the ratio of tensil stress to tensil strain
y:= s1/e1;
#                                  m (3 l + 2 m)
#                             y := -------------
#                                      l + m
# poissons ratio is the ratio of transverse strain to longitudinal strain
p:= -e2/e1;
#                                         l
#                              nu := 1/2 -----
#                                        l + m
f1:= yy= m*(3*l+2*m)/(l+m);
f2:= pp= l/(2*(l+m));
solve({f1,f2},{l,m});
#                              y p                 y
#                   {l = - --------------, m = 1/2 ------}
#                             2                    p + 1
#                          2 p  + p - 1
# let a cube of side x experience a uniform pressure s
# let dv be the change in volume
# dv is (x+dx)^3 - x^3, which is approximately 3 x^2 dx.
# then dv/v = 3 dx/ x = 3 e1 = e1 + e2 + e3
# the bulk modulus k is the ratio of the pressure s to dv/v
# for the case of uniform pressure the equations become
f1:=(l+2*m)*ee1+l*ee2+l*ee3 = s;
f2:=l*ee1 + (l+2*m)*ee2 + l*ee3 = s;
f3:= l*ee1 + l*ee2 + (l+2*m)*ee3 = s;
solve({f1,f2,f3},{ee1,ee2,ee3});
#
#                      s                s               s
#              {e3 = ---------, e2 = ---------, e1 = ---------}
#                    3 l + 2 m       3 l + 2 m       3 l + 2 m
#
e1:= s/(3*l+2*m);
```

```
k:= s/(3*e1);
#                                  k := l + 2/3 m
```

## 3.27 Cubic Polynomial

```
% cat fcubic
u2:=u*u;
u3:=u*u2;
p0:=1-3*u2 + 2*u3;
p1:=3*u2-2*u3;
p2:=u-2*u2+u3;
p3:=-u2+u3;
r:=r0*p0+r1*p1+drdu0*p2+drdu1*p3;
drdu:=diff(r,u);
dr2du2:=diff(drdu,u);
```

## 3.28 Icecream Cone Volume Difference

```
% cat icecream
# icecream volume of difference of ice cream cones
# solve equation e:=v0-2*Pi*(r1^3-r0^3)*(1-cos(t))/3=0 for r1
e:=v0-2*p*(r1^3-r0^3)*(1-cs)/3=0;
e:=subs(r0=r1-d,e);
e:=subs(p=evalf(Pi),e);
e:=expand(e);
t:=38*Pi/180;
cs:=evalf(cos(t));
e:=subs({v0=5.396179,d=.100},e);
expand(e);
solve(e,r1);
```

## 3.29 Jacobian

```
% cat minjac
f:=a*x^b*(exp(-c*x)+d)-y;
```

```
f1:=subs(x=xm,y=ym,f);
f2:=subs(x=xh,y=yh,f);
f3:=subs(x=xl,y=yl,f);
f4:=(b-c*x)*exp(-c*x)+b*d;
df1da:=diff(f1,a);
df2da:=diff(f2,a);
df3da:=diff(f3,a);
df4da:=diff(f4,a);
df1db:=diff(f1,b);
df2db:=diff(f2,b);
df3db:=diff(f3,b);
df4db:=diff(f4,b);
df1dc:=diff(f1,c);
df2dc:=diff(f2,c);
df3dc:=diff(f3,c);
df4dc:=diff(f4,c);
df1dd:=diff(f1,d);
df2dd:=diff(f2,d);
df3dd:=diff(f3,d);
df4dd:=diff(f4,d);
readlib(fortran);
fortran(df1da);
fortran(df2da);
fortran(df3da);
fortran(df4da);
fortran(df1db);
fortran(df2db);
fortran(df3db);
fortran(df4db);
fortran(df1dc);
fortran(df2dc);
fortran(df3dc);
fortran(df4dc);
fortran(df1dd);
fortran(df2dd);
fortran(df3dd);
fortran(df4dd);
```

## 3.30 Coefficients of Quadratic Equation for Intersection of Line and Cone

```
% cat maplecone
# maplecone, coefficients of quadratic equation for intersection of line and cone
 e1:= h*h*(x*x+y*y) - r*r*(z-h)^2;
e2:=subs({x=x_0+v_x*t,y=y_0+v_y*t,z=z_0+v_z*t},e1);
e3:=expand(e2);
e4:=collect(e3,t);
a:=coeff(e4,t,2);
b:=coeff(e4,t,1);
c:=coeff(e4,t,0);
a:=collect(a,h);
b:=collect(b,h);
bh:=coeff(b,h,2);
b1:=b-bh*h^2;
br:=coeff(b1,r,2);
difference:=expand(b-(bh*h^2+br*r^2));
b:= bh*h^2+br*r^2;
c:=collect(c,h);
ch:=coeff(c,h,2);
c1:=c-ch*h^2;
cr:=coeff(c1,r,2);
difference:=expand(c-(ch*h^2+cr*r^2));
c:=ch*h^2+cr*r^2;
readlib(latex);
latex(A=a);
latex(B=b);
latex(C=c);
```

## 3.31 A Recursive Program to Compute Sum of Powers Formulas

We shall give a program that computes the formula for the sum of the first $n$ integers raised to the $k$th power. First we shall derive the recursive formula.

Hint of Proof? : Courant **Differential and Integral Calculus**, V1 1934 p27. Let

$$s_k = \sum_{j=1}^{n} j^k.$$

We have

$$\sum_{j=1}^{n}[(j+1)^{k+1} - j^{k+1}] = 2^{k+1} - 1^{k+1} + 3^{k+1} - 2^{k+1} + ... + (n+1)^{k+1} - n^{k+1} =$$

$$(n+1)^{k+1} - 1.$$

On the other hand, using the binomial theorem

$$\sum_{j=1}^{n}[(j+1)^{k+1} - j^{k+1}] =$$

$$\sum_{j=1}^{n}\left[\sum_{p=0}^{k+1}\begin{pmatrix} k+1 \\ p \end{pmatrix} j^p - j^{k+1}\right] =$$

$$\sum_{j=1}^{n}\sum_{p=0}^{k}\begin{pmatrix} k+1 \\ p \end{pmatrix} j^p =$$

$$\sum_{p=0}^{k}\begin{pmatrix} k+1 \\ p \end{pmatrix} s_p(n).$$

Then

$$\sum_{p=0}^{k}\begin{pmatrix} k+1 \\ p \end{pmatrix} s_p(n) = (n+1)^{k+1} - 1.$$

So

$$\begin{pmatrix} k+1 \\ k \end{pmatrix} s_k(n) = (n+1)^{k+1} - 1 - \sum_{p=0}^{k-1}\begin{pmatrix} k+1 \\ p \end{pmatrix} s_p(n).$$

Then

$$s_k(n) = \frac{1}{k+1}\left[(n+1)^{k+1} - 1 - \sum_{p=0}^{k-1}\begin{pmatrix} k+1 \\ p \end{pmatrix} s_p(n)\right].$$

Now $s_0(n) = 1 + ... + 1 = n$, so for example the recursive formula gives after simplification

$$s_1(n) = \frac{n(n+1)}{2}.$$

We can also use the Maple sum function to do this.

```
interface(echo=2);

read(sumofpow);

# sumofpow 1^k+2^k+...+n^k  see 2/23/92 or courant
> sumofpow:= proc(k,n)
> local v,p;
> if k = 0 then v:=n fi;
> if k=1 then v:=n*(n+1)/2 fi;
> if k>1 then
>  v:= (n+1)^(k+1)-1;
>  for p from 0 to k-1 do
>   v:= v -  binomial(k+1,p)* sumofpow(p,n);
>  od;
>  v:=v/(k+1);
> fi;
> v;
> end;



sumofpow := proc(k,n)
            local v,p;
                if k = 0 then v := n fi;
                if k = 1 then v := 1/2*n*(n+1) fi;
                if 1 < k then
                    v := (n+1)^(k+1)-1;
                    for p from 0 to k-1 do
                        v := v-binomial(k+1,p)*sumofpow(p,n)
                    od;
                    v := v/(k+1)
                fi;
```

```
            v
        end
```

```
sumofpow(1,n);
```

$$\frac{1}{2}\,n\,(\,n+1\,)$$

```
sumofpow(2,n);
```

$$\frac{1}{3}\,(\,n+1\,)^3-\frac{1}{3}-\frac{1}{3}\,n-\frac{1}{2}\,n\,(\,n+1\,)$$

```
expand(");
```

$$\frac{1}{3}\,n^3+\frac{1}{2}\,n^2+\frac{1}{6}\,n$$

```
factor(");
```

$$\frac{1}{6}\,n\,(\,n+1\,)\,(\,2\,n+1\,)$$

```
factor(expand(sum(k^2,k=1..n)));
```

$$\frac{1}{6}\,n\,(\,n+1\,)\,(\,2\,n+1\,)$$

```
factor(expand(sumofpow(5,n)));
```

$$\frac{1}{12}\,n^2\,(\,2\,n^2+2\,n-1\,)\,(\,n+1\,)^2$$

```
subs(n=1000,");
```

$$167167083333250000$$

```
Sum(k^5,k=1..n) = "";
```

$$\sum_{k=1}^{n} k^5=\frac{1}{12}\,n^2\,(\,2\,n^2+2\,n-1\,)\,(\,n+1\,)^2$$

## 3.32 Unique Cubic

```
% cat readme
see maple.cnt file for contents of files
% cat uniqcubic
f:= (2*(1-z/3)+2*(z+1/3))*(1-z/3)^2+(2*(z-1/3)+tb*(1+z/3))*(1+z/3)^2;
g:=subs(tb=-2,f);
g:=normal(g);
solve(g=0,z);
f:= expand(f);
df:=diff(f,z);
k:=resultant(f,df,z);
s:=solve(k=0,tb);
evalf(s[1]);
evalf(s[2]);
evalf(s[3]);
q:=subs(tb=s[2],f);
h:=solve(q=0,z);
evalf(h[1]);
evalf(h[2]);
evalf(h[3]);
```

## 3.33 The Laplace Operator In Polar Coordinates

We shall Show that the Laplacian operator in polar (cylindical) coordinates is the Cartesian Laplacian operator. The Laplace operator in polar coordinates is

$$\frac{\partial^2 g}{\partial r^2} + \frac{1}{r}\frac{\partial g}{\partial r} + \frac{1}{r^2}\frac{\partial^2 g}{\partial \theta^2} + \frac{\partial^2 g}{\partial z^2}.$$

The $z$ partials are the same in both systems.

```
read(lapop);
```

```
# filename: lapop, Show that the Laplacian operator in cylindical
# coordinates is the Cartesian Laplacian operator.
```

```
> x=r*cos(t);
```

$$x = r\cos(t)$$

```
> y=r*sin(t);
```

$$y = r\sin(t)$$

```
> g(r,t)=f(x,y);
```

$$\mathrm{g}(r,t) = \mathrm{f}(x,y)$$

```
> d2grr:=diff(f(x(r,t),y(r,t)),r,r):
> subs({x(r,t)=r*cos(t),y(r,t)=r*sin(t)},d2grr):
> d2grr:=expand("):
> Diff(g,r,r)=d2grr;
```

$$\begin{aligned}\frac{\partial^2}{\partial r^2} g = {} & D_{1,1}(f)(r\cos(t), r\sin(t))\cos(t)^2 \\ & + 2\cos(t)\, D_{1,2}(f)(r\cos(t), r\sin(t))\sin(t) \\ & + D_{2,2}(f)(r\cos(t),\ r\sin(t))\sin(t)^2\end{aligned}$$

```
> d2gtt:=diff(f(x(r,t),y(r,t)),t,t):
> subs({x(r,t)=r*cos(t),y(r,t)=r*sin(t)},d2gtt):
> d2gtt:=expand("):
> Diff(g,t,t)=d2gtt;
```

$$\begin{aligned}\frac{\partial^2}{\partial t^2} g = {} & r^2\sin(t)^2 D_{1,1}(f)(r\cos(t), r\sin(t)) \\ & - 2r^2\sin(t)\, D_{1,2}(f)(r\cos(t), r\sin(t))\cos(t) \\ & - D_1(f)(r\cos(t), r\sin(t))\, r\cos(t) \\ & + r^2\cos(t)^2 D_{2,2}(f)(r\cos(t), r\sin(t)) \\ & - D_2(f)(r\cos(t), r\sin(t))\, r\sin(t)\end{aligned}$$

```
> dgr:=diff(f(x(r,t),y(r,t)),r):
> subs({x(r,t)=r*cos(t),y(r,t)=r*sin(t)},dgr):
> dgr:=expand("):
> Diff(g,t,t)=dgr:
> d2grr + dgr/r + d2gtt/(r*r):
> expand("):
> factor("):
> s:=simplify(",trig);
```

$$s := D_{1,1}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,) + D_{2,2}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,)$$

```
> Diff(g,r,r)+ (1/r)*Diff(g,r) + (1/(r*r))*Diff(g,t,t) = s;
```

$$\left(\frac{\partial^2}{\partial r^2}\,g\right) + \frac{\frac{\partial}{\partial r}\,g}{r} + \frac{\frac{\partial^2}{\partial t^2}\,g}{r^2} = D_{1,1}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,) + D_{2,2}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,)$$

```
> s = Diff(f,x,x) + Diff(f,y,y);
```

$$D_{1,1}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,) + D_{2,2}(\,f\,)(\,r\cos(\,t\,), r\sin(\,t\,)\,) = \left(\frac{\partial^2}{\partial x^2}\,f\right) + \left(\frac{\partial^2}{\partial y^2}\,f\right)$$

## 3.34 Contents of Files

```
% cat maple.cnt
maple.cnt    contents of maple files
maple1in
min2   compute center of conic
min3   least squares c1 x^2 + c1 x + c3 = y
min4   curl and divergence of point source
mout4  curl and divergence of point source
mout5  binomial theorem and taylor expansion
mout6  integration and diff
min18  solve least squares equation c1 yi^2 + c2 yi = xi^2 (cone)
```

# Chapter 4

# Commands

## 4.1 Plotting

```
> interface(plotdevice=postscript);
> interface(plotoutput=p.ps);
> plot(sin(x),x=0..10.);
```

## 4.2 Plotting A Bessel Function Expression

The zeroes of the following expression determine the frequences of fundamental modal vibrations of a circular clamped plate. The plot shows two zeroes approximately equal to 5.906 and 9.197.

```
f:=BesselJ(2,x)*diff(BesselI(2,x),x) - BesselI(2,x)*diff(BesselJ(2,x),x);
```

$$f := \mathrm{BesselJ}(2, x)\left(\mathrm{BesselI}(1, x) - 2\,\frac{\mathrm{BesselI}(2, x)}{x}\right) - \mathrm{BesselI}(2, x)\left(\mathrm{BesselJ}(1, x) - 2\,\frac{\mathrm{BesselJ}(2, x)}{x}\right)$$

```
plot(f,x=0. .. 10.);

 plotsetup(ps,plotoutput=‘p.ps‘,plotoptions=‘portrait,noborder‘);

plot(f,x=0. .. 8.);
```




Figure 4.1: Bessel Function Expression.

## 4.3 3D Plotting

The 3dplot may be altered by changing menu items in the plot window. Move the viewing direction with the left button. Redisplay by clicking the middle button. Printing will produce a landscape picture with border.

```
#To create a postscript file suitable for inclusion in a tex document,
#uncomment the following:
#plotsetup(ps,plotoutput='p.ps',plotoptions='portrait,noborder');
k:=11.6198;
m:=3;
plot3d([r*cos(t),r*sin(t),BesselJ(2,k*r)*cos(m*t)],
r=0 .. 1, t=0 .. 2*Pi,grid=[30,60],orientation=[60,25],style=PATCH);
```

## 4.4 Latex Information

The style files discussed below can be found in the Silicon Graphics directory

```
/apps/maple/mapleV.3/etc/
```



Figure 4.2: Vibrating Plate.

Here is a Maple text file found in that directory explaining Latex processing for maple.

```
% cat latex.txt
Printing and Viewing Maple LaTeX Documents
===========================================

To format and print documents "exported" to LaTex by Maple you will
need access to a standard version of LaTeX and the two LaTeX style
files "maplems.sty" and "mapleenv.sty" that are provided with this
document.

Place copies of these style files in the same directory (or folder) as
the file to be processed, and process the file with "latex" to create a
.dvi file.

The style file "maplems.sty" provides a complete document style based
on the LaTeX "article" style and which is suitable for presenting
typical worksheet length documents and examples.  The document style
line used in the exported worksheet is:

\documentstyle[maplems]{article}

The style file "mapleenv.sty" is read by the style file "maplems.sty"
and is used to define the various environments used in the LaTeX
document produced by Maple.  This includes definitions for the
environments

\begin{mapleinput}  ... \end{mapleinput}

\begin{maplettyout} ... \end{maplettyout}

\begin{maplelatex}  ... \end{maplelatex}

a macro used to display the separator line found between Maple regions

\maplesepline
```

```
and the macro

\mapleplot{}

facilitating the inclusion of plots (see below).

If you wish to print the Maple worksheet using a different page layout
or style, be sure to modify the chosen style to make use of
"mapleenv.sty" as in (for example)

\documentstyle[mapleenv]{book}

Changing Layout Parameters
==========================

The spacing between regions and the actual layout of the page can be
changed by modifying some of the parameters shown at the top of the
file  "maplems.sty".  To do this,

1.  Copy a version of this file to the same directory as your LaTeX
document using a different style name ( e.g., "mymaplems.sty" ) and
edit the parameters found in the file to values of your choice.

2.  Modify the document style line in your exported worksheet
appropriately.

Permission is granted to use these Maple styles without charge. It may
also be redistributed providing that they are distributed as is and
without charge.  You may modify these macros for your own use, and
redistributed subject to the following condition:

1. The style name used for any modified version of this file is
   different from mapleenv.sty.

2. You acknowledge this file as the source.
```

```
3. It is redistributed under the same terms and conditions
   as indicated here.


Including Plots
===============

The documents exported from Maple can be modified by hand to include
PostScript plots.  The macro "\mapleplot" can be inserted directly into
the latex source file.  It uses one argument: the name of the file
containing a  PostScript version of the plot.

\mapleplot{postscript.out}

Such plots are easily generated and saved to files by using Maple V.

This macro definition makes use of the style file "psfig.sty".  The
"psfig" macros are not commercially distributed, but can already be
found at many ftp sites.  It can be easily modified to rely on your
favourite macro package.

# $Id: latex.txt,v 1.2 1994/03/07 16:07:51 jsdevitt Exp $
# $Revision: 1.2 $
```

## 4.5 Substitution of a Bessel Function Into Bessel's Equation

```
diff(BesselJ(n,r),r);
```

$$-\mathrm{BesselJ}(\,n+1,r\,)+\frac{n\,\mathrm{BesselJ}(\,n,r\,)}{r}$$

```
diff(BesselJ(n,r),r,r);
```

$$-\mathrm{BesselJ}(\,n,r\,)+\frac{(\,n+1\,)\,\mathrm{BesselJ}(\,n+1,r\,)}{r}-\frac{n\,\mathrm{BesselJ}(\,n,r\,)}{r^2}$$

$$+\frac{n\left(-\mathrm{BesselJ}(n+1,r)+\frac{n\,\mathrm{BesselJ}(n,r)}{r}\right)}{r}$$

```
f(r)=BesselJ(n,r);
```

$$\mathrm{f}(r)=\mathrm{BesselJ}(n,r)$$

```
a:=(r^2 - n^2)*BesselJ(n,r);
```

$$a:=(r^2-n^2)\,\mathrm{BesselJ}(n,r)$$

```
b:=r*diff(BesselJ(n,r),r);
```

$$b:=r\left(-\mathrm{BesselJ}(n+1,r)+\frac{n\,\mathrm{BesselJ}(n,r)}{r}\right)$$

```
c:=r^2*diff(BesselJ(n,r),r,r);
```

$$c:=r^2\left(-\mathrm{BesselJ}(n,r)+\frac{(n+1)\,\mathrm{BesselJ}(n+1,r)}{r}-\frac{n\,\mathrm{BesselJ}(n,r)}{r^2}+\frac{n\left(-\mathrm{BesselJ}(n+1,r)+\frac{n\,\mathrm{BesselJ}(n,r)}{r}\right)}{r}\right)$$

```
s:=a+b+c;
```

$$s:=(r^2-n^2)\,\mathrm{BesselJ}(n,r)+r\left(-\mathrm{BesselJ}(n+1,r)+\frac{n\,\mathrm{BesselJ}(n,r)}{r}\right)+r^2\left(-\mathrm{BesselJ}(n,r)+\frac{(n+1)\,\mathrm{BesselJ}(n+1,r)}{r}-\frac{n\,\mathrm{BesselJ}(n,r)}{r^2}+\frac{n\left(-\mathrm{BesselJ}(n+1,r)+\frac{n\,\mathrm{BesselJ}(n,r)}{r}\right)}{r}\right)$$

```
expand(");
```

$$0$$